

Programming assignment extra credit 2

Hardy Jones

Professor Köppe

This program is a series of literate Haskell modules split across multiple files. So it could be run directly if that were required.

We have our three tasks:

1. The tree shortcut algorithm.
2. The greedy TSP algorithm as an optimization problem over independent systems:
3. Christofides' algorithm.

Here's an explanation of each algorithm.

1. The tree shortcut algorithm first looks at a graph, and creates a minimum spanning tree (MST). Then it walks along the MST and adds each nodes seen in preorder. The shortcut part comes from the preorder traversal. When walking the tree, If there's a path $x \rightarrow y \rightarrow z$ and y has already been seen, drop the path through y and just take the path $x \rightarrow z$. This works because if y is already seen, that means it's already in the tour. And assuming that the distance between cities is a metric, the solution is still feasible by the triangle inequality. It also imposes an upper bound on the solution of being no more than twice the optimal solution. With a proof and more information given in [1].
2. The greedy algorithm:
 - (a) Start with no edges
 - (b) Add the edge of minimum distance that does not create a short cycle or increase any vertex degree to larger than 2.
 - (c) Repeat until a full tour is formed. That's pretty much the entirety of the algorithm.
3. Christofides' algorithm:
 - (a) Find a minimum spanning tree T .
 - (b) Find a perfect matching M among vertices with odd degree.
 - (c) Combine the edges of M and T to make a multigraph G .
 - (d) Find an Euler cycle in G by skipping vertices already seen. More information is provided in [2].

Again, we first require some pragmas to let us know if something slipped by.

```
> {-# OPTIONS_GHC -Wall #-}
> {-# OPTIONS_GHC -Werror #-}
> {-# OPTIONS_GHC -fno-warn-unused-do-bind #-}
```

Now on to our actual program:

```
> module PX2 where

> import PX2.Algorithm ()

> main :: IO ()
> main = do
>     writeFile "comparison.md" ""
```

Table: Comparison of distances

Here we talk about all the algorithms we need.

```
> module PX2.Algorithm where

> import Control.Lens ((|>), (^..), _1, _2, _3, each, view)
>
> import Data.List (group, sort, sortOn)
> import Data.Tree
>
> import GHC.Natural
>
> import PX2.Graph
>
> import qualified Data.Set as S
> import qualified Data.Map as M

> treeShortcut :: (Ord v, Ord w) => Graph v w -> [v]
> treeShortcut = treeShortcut' kruskal

> treeShortcut' :: Ord v => (Graph v w -> MST v w) -> Graph v w -> [v]
> treeShortcut' f graph = walk $ f graph
```

We don't convert this to an actual tree, so the walk is a bit odd. It's possible that it's not even correct, but it seems to do the job.

```
> walk :: Ord v => MST v w -> [v]
> walk mst = go S.empty (S.toAscList . unweightedEdges $ mst)
>   where
>     go :: Ord v => S.Set v -> [(v, v)] -> [v]
>     go vs [] = []
>     go vs es@((i, j):_) = snd (go' vs i es)
>   where
>     go' vs i []
>       | S.member i vs = (vs, [])
>       | otherwise    = (S.insert i vs, [i])
>     go' vs i ((j, k):es')
>       | not (S.member i vs) =
>         (i:) <$> go' (S.insert i vs) i ((j, k):es')
>       | i == k && not (S.member j vs) =
>         let (vs', xs) = (j:) <$> go' (S.insert j vs) j es
>         in (xs ++> go' vs' i es')
>       | i == j && not (S.member k vs) =
>         let (vs', xs) = (k:) <$> go' (S.insert k vs) k es
>         in (xs ++> go' vs' i es')
>       | otherwise =
>         go' vs i es'

> kruskal :: (Ord v, Ord w) => Graph v w -> MST v w
> kruskal = go S.empty S.empty . sorted . edges
>   where
>     go :: (Ord v, Ord w) => Family v -> MST v w -> [(v, v, w)] -> MST v w
```

```

> go us a [] = a
> go us a (e@(i, j, _):es) = case unioned us i j of
>   (Just vs, Just ws)
>     | vs == ws    -> go us a es
>   (Just vs, Just ws) -> go (reunion us vs ws) (S.insert e a) es
>   (Nothing, Just ws) -> go (reunion' us ws i) (S.insert e a) es
>   (Just vs, Nothing) -> go (reunion' us vs j) (S.insert e a) es
>   (Nothing, Nothing) -> go (reunion'' us i j) (S.insert e a) es
> sorted :: Ord w => S.Set (v, v, w) -> [(v, v, w)]
> sorted = sortOn (view _3) . S.toList

> unioned :: Ord v => Family v -> v -> v -> (Maybe (S.Set v), Maybe (S.Set v))
> unioned us i j = (lookupFamily us i, lookupFamily us j)
> lookupFamily :: Ord v => Family v -> v -> Maybe (S.Set v)
> lookupFamily us u = fst <$> S.minView (S.filter (S.member u) us)
> reunion :: Ord v => Family v -> S.Set v -> S.Set v -> Family v
> reunion us vs ws = S.insert (S.union vs ws) $ S.delete ws $ S.delete vs us
> reunion' :: Ord v => Family v -> S.Set v -> v -> Family v
> reunion' us vs w = reunion us vs (S.singleton w)
> reunion'' :: Ord v => Family v -> v -> v -> Family v
> reunion'' us v w = reunion' us (S.singleton v) w

> cycles :: Ord v => Graph v w -> Bool
> cycles g = vSize g > 2 && go S.empty (S.toList $ unweightedEdges $ edges g)
>   where
>     go :: Ord v => Family v -> [(v, v)] -> Bool
>     go us [] = False
>     go us ((i, j):es) = case unioned us i j of
>       (Just vs, Just ws) | vs == ws    -> True
>       (Just vs, Just ws)                -> go (reunion us vs ws) es
>       (Just vs, Nothing)                -> go (reunion' us vs j) es
>       (Nothing, Just ws)                -> go (reunion' us ws i) es
>       (Nothing, Nothing)                -> go (reunion'' us i j) es

> incidents :: Ord v => S.Set (v, v, w) -> M.Map v Natural
> incidents = foldMap (M.singleton <$> head <*> fromIntegral . length)
>   . group
>   . sort
>   . (^..traverse.each)
>   . S.toList
>   . unweightedEdges

> tooIncident :: Ord v => S.Set (v, v, w) -> Bool
> tooIncident = tooIncident' 2

> tooIncident' :: Ord v => Natural -> S.Set (v, v, w) -> Bool
> tooIncident' n es = n < maximum (incidents es)

> greedy :: (Ord v, Ord w) => Graph v w -> [v]
> greedy g = go S.empty . sortOn (view _3) . S.toList . edges $ g
>   where
>     go ps _ | vSize g - 1 == S.size ps = walk ps

```

```

> go ps [] = walk ps
> go ps (e:es) = let ps' = S.insert e ps in
>   if cycles g {edges = ps'} || tooIncident ps'
>   then go ps es
>   else go ps' es

```

Since the graph and tree packages on hackage are pretty much terrible, we roll our own graph.

```

> {-# LANGUAGE DeriveAnyClass #-}
> {-# LANGUAGE DeriveDataTypeable #-}
> {-# LANGUAGE DeriveFoldable #-}
> {-# LANGUAGE DeriveGeneric #-}
> {-# LANGUAGE TupleSections #-}
> module PX2.Graph where

> import Control.Lens
>
> import Data.Data
> import Data.Traversable (for)
>
> import GHC.Generics
>
> import Test.QuickCheck.Arbitrary
>
> import qualified Data.Set as S

> data Graph v w = Graph
>   { vertices :: S.Set v
>   , edges    :: S.Set (v, v, w)
>   } deriving (Data, Eq, Foldable, Generic, Ord, Show, Typeable)
> type MST v w = S.Set (v, v, w)
> type Family v = S.Set (S.Set v)
>
> instance (Arbitrary v, Arbitrary w, Ord v, Ord w) => Arbitrary (Graph v w) where
>   arbitrary = do
>     vs <- S.fromList <$> arbitrary
>     let es = [(i, j) | i <- S.toList vs, j <- S.toList vs, i < j]
>     es' <- for es $ \ (i, j) -> (i, j, ) <$> arbitrary
>     pure $ Graph vs (S.fromList es')

> unweightedEdges :: Ord v => S.Set (v, v, w) -> S.Set (v, v)
> unweightedEdges =
>   S.fromList . (map $ (,) <$> view _1 <*> view _2) . S.toList

> cartesian :: (Ord a, Ord b) => S.Set a -> S.Set b -> S.Set (a, b)
> cartesian xs ys = S.fromList $ (,) <$> S.toList xs <*> S.toList ys

> vSize :: Graph v w -> Int
> vSize = S.size . vertices
> eSize :: Graph v w -> Int
> eSize = S.size . edges

```

We want to verify some facts using tests that are a bit too time consuming to verify with the type system.

```

> {-# LANGUAGE OverloadedLists #-}
> module PX2.Test where

> import Control.Lens
>
> import GHC.Natural
>
> import PX2.Algorithm
> import PX2.Graph
>
> import Test.QuickCheck
>
> import qualified Data.Set as S
> import qualified Data.Map as M

```

Generated graphs should have the right number of edges.

```

> prop_ArbitraryEdgeNumbers :: Graph Natural Natural -> Bool
> prop_ArbitraryEdgeNumbers g = (vSize g * (vSize g - 1) `div` 2) == eSize g

```

Given a graph $G = (V, E)$, E should be a subset of $V \times V$.

```

> prop_EdgesAreSubset :: Graph Natural Natural -> Bool
> prop_EdgesAreSubset g = unweightedEdges es `S.isSubsetOf` cartesian vs vs
>   where
>     vs = vertices g
>     es = edges g

```

A graph with less than three vertices cannot have cycles.

```

> prop_SmallGraphNoCycles :: Graph Natural Natural -> Property
> prop_SmallGraphNoCycles g = vSize g < 3 ==> not (cycles g)

```

Given a graph $G = (V, E)$, an MST of G should have exactly $\max(0, |V| - 1)$ edges.

```

> prop_MSTEdges :: (Graph Natural Natural -> MST Natural Natural)
>                -> Graph Natural Natural
>                -> Bool
> prop_MSTEdges f g = max 0 (vSize g - 1) == S.size (f g)

```

Given a graph $G = (V, E)$, an MST of G should be exactly V .

```

> prop_MSTVertices :: (Graph Natural Natural -> MST Natural Natural)
>                  -> Graph Natural Natural
>                  -> Property
> prop_MSTVertices f g = vSize g > 1 ==> vertices g == vs
>   where
>     vs = S.fromList $ (^..traverse.each) $ S.toList $ unweightedEdges $ f g

```

Given a graph $G = (V, E)$, a tree shortcut should be exactly V .

```

> prop_treeShortcut :: Graph Natural Natural -> Property
> prop_treeShortcut g = vSize g > 1 ==> vertices g == vs
>   where
>     vs = S.fromList $ treeShortcut g

```

Given a graph $G = (V, E)$, a walk of the MST of G should give exactly V .

```

> prop_walkVertices :: (Graph Natural Natural -> MST Natural Natural)
>   -> Graph Natural Natural
>   -> Property
> prop_walkVertices f g = vSize g > 1 ==> vertices g == vs
>   where
>     vs = S.fromList $ walk $ f g

```

Given a complete graph $G = (V, E)$, the incidents of E should have a maximum of $|V| - 1$.

```

> prop_IncidentsMaximum :: Graph Natural Natural -> Property
> prop_IncidentsMaximum g = vSize g > 1 ==>
>   maximum (incidents $ edges g) == fromIntegral (vSize g - 1)

```

Given a complete graph $G = (V, E)$, the incidents of E should be exactly V .

```

> prop_IncidentsKeys :: Graph Natural Natural -> Property
> prop_IncidentsKeys g = vSize g > 1 ==>
>   M.keysSet (incidents $ edges g) == vertices g

```

Given a graph $G = (V, E)$, a greedy G should be exactly V .

```

> prop_GreedyVertices :: Graph Natural Natural -> Property
> prop_GreedyVertices g = vSize g > 1 ==> S.fromList (greedy g) == vertices g

```

Helper for unit tests/TDD.

```

> quickAssert :: Testable prop => prop -> IO ()
> quickAssert = quickCheckWith stdArgs { maxSuccess = 1 }

> main :: IO ()
> main = do
>   quickCheck prop_ArbitraryEdgeNumbers
>   quickCheck prop_EdgesAreSubset
>   quickCheckWith stdArgs {maxDiscardRatio = 100} prop_SmallGraphNoCycles
>   quickCheck $ prop_MSTEdges kruskal
>   quickCheck $ prop_MSTVertices kruskal
>   quickCheck $ prop_walkVertices kruskal
>   quickCheck prop_treeShortcut
>   quickCheck prop_IncidentsMaximum
>   quickCheck prop_IncidentsKeys
>   quickCheck prop_GreedyVertices

```

Use some TDD to figure out the walk.

```

> quickAssert (walk (S.empty :: MST Natural Natural) == [])
> quickAssert (walk [(1,2,10)] == [1,2])
> quickAssert (walk [(1,2,10),(2,3,2)] == [1,2,3])
> quickAssert (walk [(2,3,0),(2,5,0),(5,6,0)] == [2,3,5,6])
> quickAssert (walk [(2,10,1),(10,11,2),(8,11,0)] == [2,10,11,8])

```

Use some TDD to figure out cycles.

```

> quickAssert (cycles $ Graph [1,2,3] [(1,2,0),(1,3,0),(2,3,0)])
> quickAssert (not $ cycles $ Graph [1,2,3] [(1,3,0),(2,3,0)])
> quickAssert (not $ cycles $ Graph [1..7] [(1,3,0),(2,4,0),(3,6,0),(4,5,0),(5,7,0)])
> quickAssert (not $ cycles $ Graph [1..7] [(1,2,0),(1,3,0),(2,4,0),(3,6,0),(4,5,0),(5,7,0)])
> quickAssert (cycles $ Graph [1..7] [(1,2,0),(1,3,0),(2,4,0),(3,6,0),(4,5,0),(5,7,0),(6,7,0)])

```

References

1. L. Lovász, J. Pelikán, K. Vesztergombi. *Discrete mathematics : elementary and beyond*. pp 162-163. Springer, 2003.
2. Paul E. Black, “Christofides algorithm”, in Dictionary of Algorithms and Data Structures [online], Vreda Pieterse and Paul E. Black, eds. 28 October 2008. (May 31, 2015) Available from: <http://www.nist.gov/dads/HTML/christofides.html>